

建筑结构计算机分析及程序

江见鲸 傅德炫 王立翔 编著

清华大学出版社

建筑结构计算机分析及程序

江见鲸　傅德炫　王立翔　编著

清 华 大 学 出 版 社

(京)新登字 158 号

内 容 提 要

建筑结构的内力分析和配筋计算是比较繁杂的，本书提供了编制这些计算程序的方法，包括内力计算、刚度集成、特征值计算、截面配筋、结构非线性分析和动力分析。每一段程序均有详细说明和示范源程序。

本书可供土建类大专院校师生作教学参考，也适合于广大工程技术人员阅读。

图书在版编目(CIP)数据

建筑结构计算机分析及程序/江见鲸，傅德炫，王立翔编著．—北京：清华大学出版社，1998

ISBN 7-302-02875-3

Ⅰ．建… Ⅱ．①江… ②傅… Ⅲ．①建筑结构-结构分析-计算机应用 ②建筑结构-结构计算-计算机应用 Ⅳ．TU31

中国版本图书馆 CIP 数据核字(98)第 02523 号

出版者：清华大学出版社（北京清华大学校内，邮编 100084）
　　　　因特网地址：www.tup.tsinghua.edu.cn
印刷者：北京昌平环球印刷厂
发行者：新华书店总店北京科技发行所
开　本：787×1092 1/16　印张：14.5　字数：340 千字
版　次：1998 年 7 月 第 1 版　1998 年 7 月 第 1 次印刷
书　号：ISBN 7-302-02875-3/TU・134
印　数：0001～4000
定　价：18.80 元

前　　言

本书是针对那些已经学过计算机语言(如 Fortran，Basic，C 等)和结构矩阵分析，但对编制整个程序还有一定困难的人员编写的。书中不着重公式的推导，因为公式来源在许多专业书籍中已有详细叙述，并很易找到。本书着重说明如何把计算公式变为有效的计算机程序。对于结构刚度矩阵的计算与集成，特别是集合成半带宽存储或变带宽一维存储的方法及其相应的方程组解法，都进行了较详细的说明。对于在各种受力状态下的配筋计算，其计算公式并不复杂，但各种构造要求及配筋上、下限的限制较多，程序如何适应这种需要，本书给出了必要的示例。书中对结构非线性分析及动力分析给出了详细说明。

如何把程序编得更好，更适用，更便于阅读和使用，这在开始编制程序前就应注意。一般好的程序应满足：

(1) 计算结果准确可靠，这是首先要满足的。对每一程序段均应验算以证明其可靠性，只有验算证明为准确可靠的程序，才能组合到综合程序中去。当然，综合后的程序也必须有已知正确答案的题目进行考核，以验证其可靠性。

(2) 数据输入简便，操作方便，用户界面友好。有时可用提示或图示方式引导数据输入，并编入查错功能，当输入错误或不合理的数据时会帮助纠正，对于误操作会拒绝接受。

(3) 程序结构模块化，段落分明，结构简洁，功能明确，便于程序的修改和发展。

(4) 要有足够的说明语句，不仅对数组、变量有明确说明，而且对各段程序的功能及前后联系有必要的说明。一般说明语句应占程序总语句的 1/5～1/3。这不仅便于应用程序的人员理解和使用程序，也有利于不同人员在开发大型综合程序时相互间的配合与协调发展。

在结构分析中广泛采用的有限元方法是以结点位移为基本未知量的，在杆件结构中又称为矩阵位移法。用这种位移有限元法分析结构时主要步骤如下：

(1) 输入原始数据，如控制信息、结构几何性质、材料的弹性常数、支撑约束条件等；

(2) 将结构离散化为单元，对各个单元求出其单元刚度矩阵；

(3) 将单元刚度矩阵集合成整体刚度矩阵；

(4) 计算荷载向量；

(5) 引入支撑条件，修改位移法方程；

(6) 求解位移法方程，求出结点位移；

(7) 由结点位移求出结点力或单元应变及应力；

(8) 对混凝土结构进行截面配筋计算；

(9) 输出计算结果。

本书将对其中核心部分的程序编写分章进行叙述。

江见鲸　1997 年 3 月

目　　录

1 线性代数计算

在结构静力分析中，单元刚度的计算、集成以及求解线性方程组（有限元方程组）是最花时间的。在求解结构动力问题时要计算结构的自振频率（或周期）和振型，这就涉及矩阵特征值和特征向量的计算。本章将介绍线性代数方程组的求解和矩阵特征值、特征向量的计算程序。因结构刚度矩阵是对称的，且常呈带状分布，故本章所述的方法更侧重对称矩阵，并且还介绍了矩阵元素在半带宽存储或一维压缩变带宽存储时的一些特殊算法。

1.1 高斯消去法

1.1.1 计算公式

结构静力分析时求得的有限元方程实质上是一个线性代数方程组，用矩阵形式表示为

$$[A]\{X\} = \{B\} \tag{1-1-1}$$

式中

$$[A] = \begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & \vdots & \vdots \\ a_{n1} & a_{n2} & \cdots & a_{nn} \end{bmatrix}$$

为方程组系数矩阵，即刚度矩阵，是 $n\times n$ 阶的，因刚度矩阵是对称的，故有 $a_{ij}=a_{ji}$；

$\{B\}=[b_1 \quad b_2 \quad \cdots \quad b_n]^T$——常数项列阵，即荷载向量；

$\{X\}=[x_1 \quad x_2 \quad \cdots \quad x_n]^T$——未知结点位移向量。

用高斯消去法求解线性方程组有两个过程，即消元过程和回代过程。消元过程的步骤为：

（1）首先用系数矩阵的 a_{11} 元素去除第一行的各个元素及常数项，即

$$a_{1j}^{(1)} = a_{ij}/a_{11}$$

$$b_1^{(1)} = b_1/a_{11}$$

式中上标(1)表示第一轮消元，取第一行为消元主行，也称为轴行。

（2）从第二行起，将第一行各系数和常数项乘以各行的第一列系数，并与该行相应列元素相减，这样就把除 a_{11} 外的第一列元素均变为零。

（3）从第二行开始，采用类似步骤，把轴行以下的副列元素消为零。其具体计算公式为

$$\begin{aligned} a_{kj}^{(k)} &= a_{kj}^{(k-1)}/a_{kk}^{(k-1)} \\ b_k^{(k)} &= b_k^{(k-1)}/a_{kk}^{(k-1)} \\ \left.\begin{aligned} a_{ij}^{(k)} &= a_{ij}^{(k-1)} - a_{ik}^{(k-1)}a_{kj}^{(k)} \\ b_i^{(k)} &= b_i^{(k-1)} - a_{ik}^{(k-1)}b_k^{(k)} \end{aligned}\right\} \end{aligned} \tag{1-1-2}$$

式中　$i=k+1,k+2,\cdots,n$

$j=k,k+1,\cdots,n$

消元过程结束时，原系数矩阵变为上三角矩阵，下三角(不包括主对角元素)诸元素均为零。

回代过程的步骤如下：

消元后最后一个方程为 $a_{nn}x_n=b_n$，故其解为

$$x_n = b_n/a_{nn}$$

将 x_n 代回 $n-1$ 式(倒数第二式)，可得解 x_{n-1}。逐步回代，可得所有 x_i 的解。具体计算公式为

$$x_i = \left(b_i - \sum_{j=i+1}^{n} a_{ij}\,x_j\right)\Big/a_{ii} \tag{1-1-3}$$

式中　$i=n-1,n-2,\cdots,1$

$j=i+1,i+2,\cdots,n$

在消元过程中，若 a_{kk} 特别小，用它作主元去除其它元素会带来很大的误差，甚至使计算无法进行。在这种情况下，可采用主元消去法。列主元消去法是主元消去法的一种。该法在作第 k 轮消元时，先在$[A]^{(k)}$中对 k 行以下的 k 列元素中选取绝对值最大的元素为主元，然后进行行交换，将该主元所在行换到轴行。列主元消去法程序编制不复杂，一般均可满足精度要求，因而在结构计算中应用很广。

1.1.2　计算程序

下面给出列主元消去法的计算程序SOVG，它可适用于系数矩阵不对称的情况。程序中的数组及变量说明如下：

A——$n\times n$ 二维实型数组，存系数矩阵；

B——一维数组，存常数项系数；方程组求解结束后，存放求解结果；

ES——控制计算精度的一个小数，如可取 1.0E−8。

程序SOVG清单列出如下：

```
      SUBROUTINE SOVG(N,A,B,ES,IW)
      REAL A(N,N),B(N)
      M=N-1
      DO 10 K=1,M
      C=0
      DO 5 I=K,N
         IF(ABS(A(I,K)).LE.ABS(C))GO TO 5
         C=A(I,K)
         II=I
5     CONTINUE
      IF(ABS(C).GE.ES)GO TO 15
      IW=0
      RETURN
15    IF(II.EQ.K)GO TO 20
```

```
      DO 25 J=K,N
       R=A(K,J)
       A(K,J)=A(II,J)
25    A(II,J)=R
      R=B(K)
      B(K)=B(II)
      B(II)=R
20    KP=K-1
      C=1.0/C
      B(K)=B(K)*C
      DO 10 J=KP,N
      A(K,J)=A(K,J)*C
      DO 27 I=KP,N
27    A(I,J)=A(I,J)-A(I,K)*A(K,J)
10    B(J)=B(J)-A(J,K)*B(K)
      B(N)=B(N)/A(N,N)
      DO 40 K=1,M
      I=N-K
      C=0
      IP=I+1
      DO 35 J=IP,N
35    C=C+A(I,J)*B(J)
40    B(I)=B(I)-C
      IW=1
      RETURN
      END
```

1.1.3 计算例题

已知：

$$A=\begin{bmatrix}5&7&6&5&1\\7&10&8&7&2\\6&8&10&9&3\\5&7&9&10&4\\1&2&3&4&5\end{bmatrix}\qquad B=\begin{bmatrix}96\\136\\144\\140\\60\end{bmatrix}$$

运行结果：

$$X=[3.998743\quad 4.000753\quad 4.000262\quad 3.999929\quad 3.999854]^T$$

调用子程序 SOVG 的主程序如下：

```
      REAL  A(5,5),B(5)
      DATA  A/5,7,6,5,1,7,10,8,7,2,
&            6,8,10,9,3,5,7,9,10,4,1,2,3,4,5/
      DATA B/96,136,144,140,60/
      ES=1.0E-8
      CALL SOVG(5,A,B,ES,IW)
      IF(IW.EQ.0)WRITE(*,'(A)')'Ssingular Equation'
```

```
      WRITE(*,20)(B(I),I=1,5)
20    FORMAT(2X,5F10.6)
      STOP
      END
```

1.2 高斯-亚当消去法

1.2.1 计算公式

高斯-亚当消去法是对高斯消去法的一种修正。高斯消去法在进行第 k 轮消元时，只对 k 行以下的第 k 列元素进行消去；在高斯-亚当消去法中，不仅对 k 行以下的元素进行消元，同时把 k 行以上的第 k 列元素进行消去。因而在高斯消去法消元过程结束时，系数矩阵化为上三角矩阵，然后由回代过程求得方程的解；而在高斯-亚当消去法中，在消元过程结束时，原系数矩阵已成为对角元素为 1、其余元素为零的单位矩阵。显然，在高斯-亚当消去法中不需要回代，消元结束后，方程右端原常数项已化为方程组的解。

高斯-亚当消去法的具体计算公式为

$$
\begin{aligned}
a_{kj}^{(k)} &= a_{kj}^{(k-1)}/a_{kk}^{(k-1)} \qquad j = k, k+1, \cdots, n \\
a_{ij}^{(k)} &= a_{ij}^{(k-1)} - a_{ik}^{(k-1)} a_{kj}^{(k)}
\end{aligned}
\tag{1-2-1}
$$

式中 $j=k+1,\ k+2,\ \cdots,\ n$

$i=1,2,\cdots,k-1,k+1,\cdots,n,i\neq k$

$$b_i^{(k)} = b_i^{(k-1)} - a_{ik}^{(k-1)} b_k^{(k)}$$

若右端常数项不是一列，则也可同时求得其解，如为 m 列，则对常数项运算时取

$$b_{ij}^{(k)} = b_{ij}^{(k-1)} - a_{ik}^{(k-1)} b_{kj}^{(k)} \tag{1-2-2}$$

式中 $i=1,2,\cdots,n \quad i\neq k$

$j=1,2,\cdots,m$

消元化简后的方程化为

$$
\left.
\begin{aligned}
x_i &= b_i^{(n)} \\
x_{ij} &= b_{ij}^{(n)}
\end{aligned}
\right\}
\tag{1-2-3}
$$

或

式中 $i=1,2,\cdots,n$

$j=1,2,\cdots,m$

若常数项为 n 阶的单位矩阵，则消元后的系数矩阵就化为原矩阵的逆。

1.2.2 计算程序

按上述公式，编制求解程序 SOVY，其中变量及数组的含义为：

N——整型变量，方程组的阶数；

M——整型变量，方程右端常数项；

LA——指示变量 LA=0，求矩阵的逆；LA=1，求方程的解；

A(N,N)——二维数组，存放方程的系数；

B (N,M)——二维数组，存放方程常数项；若 LA=0（即求矩阵的逆）时，B 应为

$N \times N$阶的。

```
      SUBROUTINE SOVY(A,B,N,M,LA,EP)
      REAL * 4 A(N,N),B(N,M)
      IF(LA.NE.0)GO TO 30
      DO 80 I=1,N
        DO 60 J=1,N
60      B(I,J)=0
80    B(I,I)=1.0
30    DO 90 K=1,N
        P=0.0
        DO 65 J=K,N
          IF(ABS(A(J,K)).LE.ABS(P))GO TO 65
          P=A(J,K)
          JO=J
65      CONTINUE
        IF(ABS(P).GT.EP)GO TO 40
        WRITE(*,'(A)')'  SINGULAR EQUATION'
        RETURN
40      IF(JO.EQ.K)GO TO 70
        DO 50 I=K,N
          T=A(K,I)
          A(K,I)=A(JO,I)
50      A(JO,I)=T
        DO 55 IM=1,M
          T=B(K,IM)
          B(K,IM)=B(JO,IM)
55      B(JO,IM)=T
70      P=1.0/P
        N1=N-1
        IF(K.EQ.N)GO TO 600
        DO 85 I=K,N1
          A(K,I+1)=P*A(K,I+1)
          DO 85 J=K,N1
85      A(J+1,I+1)=A(J+1,I+1)-A(J+1,K)*A(K,I+1)
600     DO 63 IM=1,M
63      B(K,IM)=B(K,IM)*P
        IF(K.EQ.N)GO TO 700
        DO 66 J=K,N1
        DO 66 IM=1,M
66      B(J+1,IM)=B(J+1,IM)-B(K,IM)*A(J+1,K)
90    CONTINUE
700   DO 74 I1=2,N
        I=N+1-I1
        DO 74 J=I,N1
          DO 74 IM=1,M
74    B(I,IM)=B(I,IM)-A(I,J+1)*B(J+1,IM)
      IF(LA.EQ.0)GO TO 800
```

```
      DO 88 I=1,M
      WRITE(*,'(/1X,A,I4/(4X,5F10.5))')
     $ 'EQUATIONS=',I,(B(J,I),J=1,N)
88    CONTINUE
      GO TO 888
800   WRITE(*,'(/1X,A)')'INVERSE OF MATRIX IS'
      DO 95 I=1,N
        WRITE(*,'(1X,5F10.4)')(B(I,J),J=1,N)
95    CONTINUE
888   RETURN
      END
```

1.2.3 计算例题

已知：$A=\begin{bmatrix}1 & -1 & 1\\ 5 & -4 & 3\\ 2 & 1 & 1\end{bmatrix}$ $B=\begin{bmatrix}-4 & -4\\ -12 & 2\\ 11 & -2\end{bmatrix}$

求：方程组的解

求解结果：

$$x_1=[3 \quad 6 \quad -1]^T$$
$$x_2=[6 \quad -2 \quad -12]^T$$

调用 SOVY 的主程序为：

```
      DIMENSION A(3,3),B(3,2)
      DATA (A(I,1),I=1,3)/1,5,2/,(A(I,2),I=1,3)/-1,-4,1/,
     3      (A(I,3),I=1,3)/1,3,1/
      DATA (B(I,1),I=1,3)/-4,-12,11/,(B(I,2),I=1,3)/-4,2,-2/
      EP=1.0E-8
      CALL SOVY(A,B,3,2,1,EP)
      STOP
      END
```

1.3 半带宽存储消去法

1.3.1 计算公式

结构刚度矩阵即有限元方程组的系数矩阵，其元素常集中在主对角元附近，呈带状分布。从对角元素到最外边的第一个非零元素的个数，称为半带宽。各行的半带宽一般不相同，各行的半带宽中选出最大的一个称为最大半带宽。显然，最大半带宽以外的元素均为零。因在消元过程中零元素可以跳过去，因而可不予存储。又因矩阵对称，可以只存对角元以上的半带宽以内的元素，称为半带宽存储。当然也可只存储下三角半带宽内的元素。本书程序按上三角半带宽存储方式编制。

刚度矩阵按半带宽存储，而一般消去法的公式是按满存时推导的，为此要找出同一元素在两种存储方式下的一一对应关系。



图 1-1

由图 1-1 可知，在全存中的某一个元素 a_{ij}(i 行、j 列元素)与半带存储中的 a_{IJ}(I 行，J 列元素)有如下关系：全存中的 i 行元素在半带存储中处于同一行，即有 $I=i$；全存中的 j 列元素在半带存储中变为斜角线上的元素，易于推知 $J=j-I+1$。于是同一元素在全存矩阵中的位置(i,j)与在半带存储中的位置(I,J)有如下关系：

$$I = i$$

$$J = j - I + 1$$

于是，

$$a_{IJ} = a_{i,j-I+1} \tag{1-3-1}$$

对于消元公式，由式(1-1-2)的第一式得

$$a_{ij}^{(k)} = a_{ij}^{(k-1)} - \frac{a_{kj}^{(k-1)}}{a_{kk}^{(k-1)}} a_{ik}^{(k-1)} \tag{1-3-2}$$

因矩阵对称，且只存上三角矩阵，利用对称性，$a_{ik}^{(k-1)}=a_{ki}^{(k-1)}$，$a_{ij}^{(k)}=a_{ji}^{(k)}$，对于第 k 轮消元，有 $i>k$。为了使消元过程在上三角部分进行，将下三角部分元素 a_{ik}换用上三角相对称的元素 a_{ki}，同时规定列码 j 大于行码 i，则第 k 轮消元公式可改写为(为书写简洁，略去前一轮上标 $k-1$)

$$\left.\begin{aligned} a_{ij}^{(k)} &= a_{ij} - \frac{a_{kj}}{a_{kk}} a_{ki} \\ b_i^{(k)} &= b_i - \frac{a_{ki}}{a_{kk}} b_k \end{aligned}\right\} \tag{1-3-3}$$

式中　$k=1,2,\cdots,n-1$

$i=k+1,k+2,\cdots,n$

$j=i,i+1,\cdots,n$

对于带存，则应利用全存与带存的对应关系。由式(1-2-1)消元公式可知，每一次计算涉及 4 个元素 a_{ij}，a_{kj}，a_{kk}，a_{ki}，其位置的对应关系见图 1-2。由式(1-3-1)的对应关系，可以确定在带存中的相应元素：

$$a_{kk} \to a_{k1}$$
$$a_{ij} \to a_{iJ} \quad J = j - i + 1$$
$$a_{ki} \to a_{kL} \quad L = i - k + 1$$
$$a_{kj} \to a_{kM} \quad M = j - k + 1 = J + i - k$$

于是，消元公式可改写为

$$\left.\begin{aligned} a_{iJ}^{(k)} &= a_{iJ} - \frac{a_{kL}}{a_{k1}} a_{kM} \\ b_i^{(k)} &= b_i - \frac{a_{kL}}{a_{k1}} b_k \end{aligned}\right\} \tag{1-3-4}$$



图 1-2

除元素下标变化外，对于行码 i 及列码 J 的变化范围也有所不同。设矩阵半带宽为 NW，一般为 N 阶。则由图 1-2(b)可见，对于第 k 轮消元，行码 i 从 $k+1$ 到 i_m，一般 $i_m=k+NW-1$，但到最后 n 行时，i_m 不能大于 N，故 $i_m=\min(k+NW-1,N)$。对于列码 J 的变化范围，则从 1 到 $NW-(i-k)=NW-(L-1)$。于是，带消去法的消元公式为

$$\left.\begin{aligned} a_{iJ}^{(k)} &= a_{iJ} - \frac{a_{kL}}{a_{k1}} a_{kM} \\ b_i^{(k)} &= b_i - \frac{a_{kL}}{a_{k1}} b_k \end{aligned}\right\} \tag{1-3-5}$$

式中 $k=1,2,\cdots,N$

$i=k+1,k+2,\cdots,i_m \quad (i_m \leqslant N)$

$J=1,2,\cdots,NW-L+1$

对于回代公式，由于 a_{ii} 与 a_{ik} 对应，a_{ij} 与 a_{iJ} 对应，故式(1-1-3)可改写为

$$\begin{aligned} x_i &= \left(b_i - \sum a_{iJ} x_j\right) \Big/ a_{i1} \\ &= \left(b_i - \sum a_{iJ} x_{J+i-1}\right) \Big/ a_{i1} \end{aligned} \tag{1-3-6}$$

由图 1-2(c)可见，当行码 i 较小时，该行最大列码为

$$J_m = NW \tag{1-3-7a}$$

当行码较大时，该行的最大列码为

$$J_m = N - i + 1 \tag{1-3-7b}$$

最大列码应取这两者中的较小值。

这样，可得消去法中的回代公式为

$$\left.\begin{aligned} x_n &= b_n / a_{n1} \\ x_i &= \left(b_i - \sum a_{iJ} x_{J+i-1}\right) \Big/ a_{i1} \end{aligned}\right\} \tag{1-3-8}$$

式中　$i=n-1, n-2, \cdots, 1$

$J=2,3,\cdots,J_m \quad (J_m \leqslant NW)$

1.3.2　计算程序

根据上述公式编出带消去法程序 SOVB，程序中各变量的含义为：

A(N,NW)——二维数组，存放方程系数；

B(N)——存放方程组右端已知向量，运行结束后存放方程组的解；

N——方程阶数；

NW——最大半带宽；

EP——控制精度用的一个小数，如取 EP=1.0E−8。

```
      SUBROUTINE SOVB(N,NW,A,B,EP)
      DIMENSION A(N,NW),B(N)
      DO 10 M=1,N
        IF(ABS(A(M,1)).LE.EP)GO TO 100
        I=M
        DO 20 L=2,NW
        I=I+1
        IF(ABS(A(M,L)).LE.EP)GO TO 20
        T=A(M,L)/A(M,1)
        J=0
        DO 30 K=L,NW
        J=J+1
30      A(I,J)=A(I,J)-T*A(M,K)
        A(M,L)=T
        B(I)=B(I)-T*B(M)
20      CONTINUE
10    B(M)=B(M)/A(M,1)
      DO 40 K1=2,N
        II=N-K1+1
        DO 40 K2=2,NW
        JJ=II+K2-1
40    B(II)=B(II)-A(II,K2)*B(JJ)
      RETURN
100   WRITE(*,'(A)')        'Singular Equation'
      RETURN
      END
```

1.3.3 计算例题

计算下列代数方程组的解，$N=4$　　$NW=2$：

$$\begin{bmatrix} 5 & 6 & 0 & 0 \\ & 5 & 6 & 0 \\ \text{对称} & & 5 & 6 \\ & & & 5 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \\ x_3 \\ x_4 \end{bmatrix} = \begin{bmatrix} 11 \\ 17 \\ 17 \\ 11 \end{bmatrix}$$

编一段主程序，调用 SOVB，取 EP=1.0E−8：

```
REAL A(4,2),B(4)
DATA A/4*5,3*6,0/
DATA B/11,17,17,11/
EP=1.0E-8
CALL SOVB(4,2,A,B,EP)
WRITE(*,'(4F10.7)')B
STOP
END
```

计算结果：$x=[1.0\quad 1.0\quad 1.0\quad 1.0]$

1.4　带存平方根法

1.4.1 计算公式

若方程组系数矩阵为对称正定的，则还可作以下分解：

$$[A]=[L][L]^{\mathrm{T}}$$

其中$[L]$为下三角矩阵，这种分解又称为乔列斯基分解。分解后原方程化为

$$[L][L]^{\mathrm{T}}\{X\}=\{B\}$$

令
$$[L]^{\mathrm{T}}\{X\}=\{Y\}$$

则
$$[L]\{Y\}=\{B\}$$

因$[L]$为下三角矩阵，很易由$\{B\}$求得$\{Y\}$；进而由$\{Y\}$求得方程的解$\{X\}$。

将矩阵展开，

$$[A]=\begin{bmatrix} a_{11} & a_{12} & \cdots & a_{1n} \\ a_{21} & a_{22} & \cdots & a_{2n} \\ \vdots & \vdots & & \vdots \\ a_{n1} & a_{n2} & \cdots & a_{nn} \end{bmatrix} = \begin{bmatrix} l_{11} & & & \\ l_{21} & l_{22} & & \\ \vdots & \vdots & \ddots & \\ l_{n1} & l_{n2} & \cdots & l_{nn} \end{bmatrix} \begin{bmatrix} l_{11} & l_{12} & \cdots & l_{1n} \\ & l_{22} & \cdots & l_{2n} \\ & & \ddots & \vdots \\ & & & l_{nn} \end{bmatrix}$$

将$[L][L]^{\mathrm{T}}$相乘的元素与$[A]$中的元素相比较，可以得到l_{ij}的计算公式。例如，对第一列，显然有

$$l_{11}=\sqrt{a_{11}}$$

$$l_{i1}=\frac{a_{i1}}{l_{11}}\qquad i=2,3,\cdots,n \tag{1-4-1}$$

进而可得

$$l_{ii} = \left(a_{ii} - \sum_{k=1}^{i-1} l_{ki}^2\right)^{\frac{1}{2}} \qquad i = 2,3,\cdots,n$$

$$l_{ij} = \left(a_{ij} - \sum_{k=1}^{i-1} l_{ki} l_{kj}\right) \Big/ l_{ii} \qquad j > i \tag{1-4-2}$$

由上公式可以按下列步骤求解方程组：

(1) $l_{11} = \sqrt{a_{11}}$；$l_{i1} = \dfrac{a_{i1}}{l_{11}}$ $\qquad i = 2,3,\cdots,n$

(2) 对于 $j = 2,3,\cdots,n$

$$l_{jj} = \left(a_{jj} - \sum_{k=1}^{j-1} l_{jk}^2\right)^{\frac{1}{2}}$$

$$l_{ij} = \left(a_{ij} - \sum_{k=1}^{j-1} l_{ik} l_{jk}\right) \Big/ l_{jj} \qquad i = j+1, j+2, \cdots, n$$

(3) 求解$[L]\{Y\} = \{B\}$

$$y_1 = b_1 / l_{11}$$

$$y_i = \left(b_i - \sum_{k=1}^{i-1} l_{ik} y_k\right) \Big/ l_{ii} \qquad i = 2,3,\cdots,n \tag{1-4-3}$$

(4) 求解$[L]^{\mathrm{T}}\{X\} = \{Y\}$

$$x_n = y_n / l_{nn}$$

$$x_i = \left(y_i - \sum_{k=i+1}^{n} l_{ki} x_k\right) \Big/ l_{ii} \qquad i = n-1, \cdots, 2, 1 \tag{1-4-4}$$

由于平方根法中系数矩阵的分解只涉及带宽内的元素，带宽外的零元素在分解后仍为零元素，因而分解后的[L]矩阵仍呈带状分布。下面给出平方根法解线性方程组的程序，其系数矩阵是按半带宽存储的，包括分解(DECOMP)及求解(SOLVE)两个子程序，可同时解 M 组常数向量。

1.4.2 计算程序

由上述公式，可编出计算机程序。程序中的变量及数组的含义说明如下：

A——$N \times NB$ 二维数组，存放方程组系数；

N——方程的阶数；

NB——方程组系数矩阵的半带宽；

B——$N \times M$ 二维数组，存放方程组右边的常数项向量，共有 M 列；

DF——一维数组，存中间变量。

```
      SUBROUTINE DECOMP(N,NB,A)
      DIMENSION A(N,NB)
      DOUBLE PRECISION DF
      A(1,1)=SQRT(A(1,1))
      DO 5 K=2,NB
5     A(1,K)=A(1,K)/A(1,1)
      DO 25 K=2,N
```